Using the WRDS System

June, 2003

1.) Create an account
Direct your web browser to http://wrds.wharton.upenn.edu/ and click  account request. Next, click  request a user account  and fill in the appropriate fields. After you submit your request, your local WRDS representative must then approve or reject the request. After the request is approved, an e-mail will be sent to you with instructions on how to set the account password to something of your choosing. If you forget your password or want to change it, you can do so by going to the same URL as above. Click members login, select either  Forgot Password  or  Change Password  and follow the instructions.

2.) Using the web interface
Direct your web browser to http://wrds.wharton.upenn.edu and click  members login. Login and select the database vendor you are interested in. Select the database you want to query from the list in the left-hand column. Instructions are provided on each web query page, but in general you will need to select a date range, a search variable (e.g., TICKER) and one or more output variables. Note that different output formats are available and the output can be compressed (in Step Four). The output will be available in a separate web browser window, and it can be saved to your local computer by following the  Download Instructions  on that page.

Note that the size of the output file from web-based queries is limited to 2 GB. If you choose a compression option, however, you can typically achieve a compression ratio of  8-to-1 or more. In practice, then, the output file size limitation is approximately 16 GB (in uncompressed, ASCII form) and is very rarely binding if one of the compression options is selected. A compressed output file can then be downloaded to your local computer for analysis.

Extensive documentation is available on the WRDS web site for each database in particular and for the use of the WRDS system in general. To get information on the use of WRDS in general, click on the  support  tab in the upper-right corner of the web page. Specific database information can be obtained within the web page for each database. The  Documentation  hyperlink will direct you to a page containing variable names and definitions. The Data Manuals  hyperlink will take you to a page containing the documentation made available by the data vendor. 

3.) Using SSH to connect to the WRDS server 
If the web interface doesn’t provide the flexibility or sophistication your analysis requires, you can login directly to the WRDS server to run your own programs on the data.  For instructions on obtaining and setting up an ssh client (and “sftp” – secure ftp) on your computer, see:

http://wrds.wharton.upenn.edu/support/docs/wrdsSSH.shtml
Once you’ve connected to the WRDS server, you can run SAS programs at the command line by entering, e.g., “sas  myfile.sas”. (Assuming the file “myfile.sas” contains SAS source code. The “log” file would then be named “myfile.log” and the “listing” file would be named “myfile.lst”.) If you’re not familiar with the UNIX operating system, see the document:


http://wrds.wharton.upenn.edu/support/docs/UNIXIntro.pdf
4.) Using X Windows to connect to the WRDS server
To use X Windows with the WRDS Unix server, you will need an X Windows and an SSH (“secure shell”)  application for your PC. The objective is to “tunnel” the X Windows information from the WRDS server to your PC via a secure (i.e., encrypted) SSH connection. Many applications are available and suitable for this purpose, but it’s not possible for WRDS to provide support for every feasible combination of applications. We do have experience, however, with Exceed (http://www.hummingbird.com) and SSH Secure Shell for Workstations (http://www.ssh.com). The following instructions describe the setup procedure for these applications, and may also provide guidance on how to set up X Windows / SSH tunneling using other applications. Your local computer support organization will undoubtedly be able to assist you further, if necessary.

a.) Setup Exceed: 

     In Xconfig, choose Communication | Mode: Passive. 

     In Xconfig, choose Window Mode | Multiple. 

b.) Setup SSH Secure Shell : 

     Under File | Profiles, select “Add Profile” (provide a name) 

     Under File | Profiles, select “Edit Profile” (select name): 

     Enter the Host Name: wrds.wharton.upenn.edu 

     Enter your wrds username. 

     Select the “Tunneling” tab: 

     Select “Tunnel X11 connections” 

c.) Start Exceed (it will be minimized on your task bar, waiting for connections). 

d.) Login to wrds.wharton.upenn.edu using SSH (select Profiles | profile name)

Within SSH, you can now get an X Windows terminal window to display on your MS Windows desktop by entering "xterm". (To put the xterm process in the background – which will return the command prompt to you – enter “xterm &”.) Other X Windows applications such as “emacs” can be displayed locally as well. 

As of this writing (Feb, 2003) it is still possible to use Exceed to connect to the WRDS server using the “X Display Manager” (XDM). This may change in the near future, however, due to security issues (sending unencrypted passwords over the network). In the meantime, you can use this alternative X-Windows interface by setting up Exceed as follows:

a.) In Xconfig, choose Communication | Mode: | XDMCP: Broadcast. Click “Configure”, check the “Host List File” box and then click “Edit”. Enter the WRDS server name, wrds.wharton.upenn.edu, in the Host List file and then click “File”, “Save” and then “File”, “Exit”. Click “OK”, “OK” and then double-click “Security”. In the “Host Access Control List” area, select “File” and then click the “Edit” button. Again, enter the WRDS server name in this file, save it and exit. Click “OK” and then double-click “Window Mode”. In the “Window Mode” area, select “Single”. (As a personal preference, I also de-select the “Close Warning on Exit” option, since it’s redundant.) Click “OK” and close the “Xconfig” window. When you start Exceed in this mode, you will be presented with a Host List – select the WRDS server from the list, click “OK” and login when prompted. Once you’ve logged-in, you will see a CDE (Common Desktop Environment) toolbar at the bottom. Click the “up-arrow” icon that is the 3rd one from the right, at the top edge of the toolbar. A menu will pop up. Right-click the 2nd icon from the bottom – the one that looks like a personal computer. Select “Promote to Front Panel”. You can now click the PC icon on the toolbar whenever you want to open a new terminal window. These terminal windows can be resized and moved around your CDE desktop as you like. 

Note that, using X Windows, it is possible to run SAS interactively on the WRDS server (by simply entering “sas” at the command prompt) with the SAS graphical interface displayed on your PC. You will almost certainly not want to use this method to interface with SAS, however, since the speed and responsiveness of the user interface will be extremely slow. PC SAS allows you to use a local (PC-based), graphical interface to SAS and make use of the WRDS server’s resources, however. (See the discussion below for more information on using PC SAS.) 

To run a SAS program on the WRDS server within a terminal window (using either ssh or Exceed), simply edit a SAS program file on the server (say, “myfile.sas”) and run the program by entering, e.g., “sas myfile” at the command prompt. The “log” file will be named “myfile.log” while the listing file will be named “myfile.lst”.

5.) Editors on the WRDS server
The WRDS server has a several editors available for you to use, including “pico”, “emacs” and other standard UNIX editors. An X-Windows editor that I personally like is “nedit”. To install this editor in your home directory, execute the following commands at the command prompt:


cd


cp  /home/wharton/robinsnd/nedit-5.3/nedit-5.3-solaris-sparc.tar.gz   ~


gunzip  nedit-5.3-solaris-sparc.tar.gz


tar  xvf  nedit-5.3-solaris-sparc.tar

Next, edit your “.cshrc” file in your home directory by entering “pico .cshrc”. (Note the leading period in the file name “.cshrc”.) Enter the following line at the bottom of the file:


alias  nedit  ~/nedit-5.3/nedit

Save the file by entering <Ctrl-x>, “y”, <Enter>. To make these changes effective in this terminal window, enter “source .cshrc”. (The next time you open a terminal window, this command will not be necessary, however.) You can now use this X-Windows editor by using the “nedit” command.

6.) Navigating the WRDS server
Each user has a “home” directory (e.g., /home/upenn/robinson) and a “projects” directory (e.g., /projects/upenn/robinson). There are disk quota limitations of 25 MB and 250 MB on each user’s “home” and “projects” directory, respectively. In addition, users are free to make use of one of 6 temporary file systems, /sastemp1 - /sastemp6, each of which has many GB of available space. To use one of these temporary file systems, just create your own directory in, say, /sastemp1. Be aware, however, that old files are automatically deleted from the temporary storage areas after 48 hrs.

All databases supported by WRDS are put in SAS data set form. In addition, CRSP and Compustat data files are made available in their original binary form for users that want to read the data with Fortran programs. Each database is contained in its own directory under the “/wrds” directory. For example, CRSP files are located in /wrds/crsp and Compustat files are located in /wrds/compustat. Each database directory also contains subdirectories named “sasdata”, “seqdata” and “samples”. The data in SAS format is contained in the “sasdata” directory, while the raw binary data (if available) will be in “seqdata”. Sample programs (in SAS and Fortran, if applicable) will be in the “samples” directory – for example, CRSP sample programs can be found in /wrds/crsp/samples. The sample programs can be copied to one of your own directories, edited to suit your needs, and run on either the WRDS server or your own computer (if appropriately modified).

7.) PC SAS  (The SAS System for Windows, ver. 8)
You can also perform analyses of WRDS-supported databases with PC SAS. The most straightforward way to do this is to “sftp” a SAS data set and a corresponding SAS sample program from the WRDS server to your PC. (See section 3, “Using SSH to connect to the WRDS server” for information on setting up sftp on your personal computer.) SAS data sets on the WRDS UNIX server are binary-compatible with PC SAS ver. 8, so there is no need to convert SAS data sets to transport format before transfer. Note that, in this case the “libname” statement in the sample program must be edited to point to the local (PC) directory containing the SAS data set.

8.) PC SAS/CONNECT – Remote Library Services (RLS)
Please note!  See the following document for current information on how to setup PC SAS/CONNECT to interact WRDS:


http://wrds.wharton.upenn.edu/support/docs/wrdsSASConnect.shtml
PC SAS/CONNECT can read a SAS data set from the WRDS server dynamically (i.e., during program execution), without first downloading it to your PC (prior to program execution). This capability is called Remote Library Services (RLS), and it allows you to run a SAS program locally (on your PC), using data that is downloaded from the WRDS server during program execution. 

Note, however, that a WHERE clause or a PROC CONTENTS that is operating on a remote SAS data set will execute on the server, not locally. In these cases the remote data set is not downloaded to the local machine before execution. See the examples below for  illustrations.

RLS may be a convenient technique for the one-time analysis of a data set that you don’t want to store locally, but it may be more efficient to download the data to your PC first, as described in section 7. For example, if there is a bug in your program and you have to run it again, the entire data set may have to be download from the server to your PC again when you run the corrected program. An example of an RLS program follows:

%let wrds = wrds.wharton.upenn.edu 4016;
options comamid=TCP remote=WRDS;
signon username=_prompt_;

options linesize=72 nocenter nodate; 
libname crsplib remote '/wrds/crsp/sasdata' server=wrds;

proc contents data=crsplib.msi;

run;

signoff;

The first 3 lines of the above program are necessary to establish a connection between the local SAS session and the remote SAS session (on the WRDS server). The libname statement defines a remote library in your local SAS Session. (Click the “Explorer” tab at the bottom of the left-hand pane in PC SAS and then double-click “Libraries” to see your currently-defined libraries.) Since this simple program only runs a PROC CONTENTS, the remote data set is not downloaded to the PC before execution. The “signoff” command terminates the link between the local SAS session and the remote SAS session. (This will remove the remote library definition in the local SAS session.)

Note that if you don’t execute the “signoff” command in your program, the link between the local and remote SAS sessions remains and you can execute additional lines of code without re-defining your remote library definition. For example, if you submit the above lines of code without the “signoff”command, you could then submit the following:


proc contents data=crsp.msf;


run;

(This can be done by putting the above lines in the Editor window of PC SAS, highlighting the lines of code, and hitting the F8 key.) Also note that the remote library definition “crsplib” remains in the “Explorer” pane of PC SAS after the program has completed.

To see that the WHERE clause operating on a remote data set is executed on the server, whereas an IF clause operating on the same data set requires the data to be download to the local machine for execution, compare the following two .log files:

78   data temp;

79   set crsplib.msf;

80   where permno = 12490 and date >= '01JAN2001'd;

81   run;

NOTE: There were 12 observations read from the data set CRSPLIB.MSF.

      WHERE (permno=12490) and (date>='01JAN2001'D);

NOTE: The data set WORK.TEMP has 12 observations and 19 variables.

NOTE: DATA statement used:

      real time           0.14 seconds

      cpu time            0.04 seconds

********************************************************************

86   data temp;

87   set crsplib.msf;

88   if permno = 12490 and date >= '01JAN2001'd;

89   run;

NOTE: There were 2981941 observations read from the data set

      CRSPLIB.MSF.

NOTE: The data set WORK.TEMP has 12 observations and 19 variables.

NOTE: DATA statement used:

      real time           9:59.08

      cpu time            28.34 seconds
Note that the time of execution is either .14 seconds or almost 10 minutes, depending on whether you use “WHERE” or “IF”. (See below for a comparison of the time of execution for the same program executed on the server using the RSUBMIT command.)

It is possible to use local libraries and remote libraries at the same time. This allows you to use SAS data sets on the WRDS server and SAS data sets on your local machine within the same program. For example:

%let wrds = wrds.wharton.upenn.edu 4016;

options comamid=TCP remote=WRDS;

signon username=_prompt_;

options linesize=72 nocenter nodate; 

libname crsplib remote '/wrds/crsp/sasdata' server=wrds;

libname locallib 'c:\sas';

data a;

set crsplib.msi;

where '01JAN2000'd <= date <= '31DEC2000'd;

run;

data b;

set locallib.msi_local;

where '01JAN2001'd <= date <= '31DEC2001'd;

run;

data c;

set a b;

run;

proc print data=c;

 var date vwretd ewretd;

run;


signoff;

9.) SAS CONNECT – Remote Submit
Since using Remote Library Services may mean that your program will need to download sizable files from the server to your local machine during program execution, it may be better to use Remote Submit to run your SAS program directly on the WRDS server. Only the log and listing files would be returned to your local PC SAS session in this case. For example:

%let wrds = wrds.wharton.upenn.edu 4016;

options comamid=TCP remote=WRDS;

signon username=_prompt_;


rsubmit;

options linesize=72 nocenter nodate; 

libname remlib '/wrds/crsp/sasdata’;

data temp;

set remlib.msf;

where permno = 12490 and date gt ‘01JAN2001’d;
run;


proc print data=temp noobs;


 var permno date ret prc;


run;


endrsubmit;


signoff;

It is useful at this point to note that using a WHERE clause on an indexed variable will always be more efficient than using an IF clause. This is because the WHERE clause makes use of the index, but the IF clause does not. To illustrate, compare the following two .log files, both of which were executed using Remote Submit:

4    data temp;

5    set remlib.msf;

6    where permno = 12490 and date gt '01JAN2001'd;

7    run;

NOTE: There were 12 observations read from the data set REMLIB.MSF.

      WHERE (permno=12490) and (date>'01JAN2001'D);

NOTE: The data set WORK.TEMP has 12 observations and 19 variables.

NOTE: DATA statement used:

      real time           0.22 seconds

      cpu time            0.07 seconds

4    data temp;

5    set remlib.msf;

6    if permno = 12490 and date gt '01JAN2001'd;

7    run;

NOTE: There were 2981941 observations read from the data set REMLIB.MSF.

NOTE: The data set WORK.TEMP has 12 observations and 19 variables.

NOTE: DATA statement used:

      real time           41.79 seconds

      cpu time            37.53 seconds

The IF clause took significantly longer than the WHERE clause - 42 seconds vs. .22 seconds. Of course, for larger and more complex queries the difference will amount to a great deal of time. So, use the WHERE clause, if possible.

10.) PC SAS/CONNECT - PROC UPLOAD / DOWNLOAD

When using PC SAS/CONNECT, it is possible to move files between your PC and the UNIX server during program execution. The SAS Procedures PROC UPLOAD and PROC DOWNLOAD will upload (from your PC to the UNIX server) or download (from the UNIX server to your PC) files during program execution. This technique is useful when you have a local file (on your PC) containing company identifiers and dates, and you want to use it in a SAS program that you will Remote Submit to run on the UNIX server. Once the program has extracted the desired subset of data, it can automatically be downloaded to your PC at the end of the program.  For example:

%let wrds = wrds.wharton.upenn.edu 4016;

options comamid=TCP remote=WRDS;

/* ----------------------------------------------------- */

/* Note that fileref and libref statements that refer to */

/* files and libraries on the local machine (PC) MUST BE */

/* EXECUTED PRIOR TO THE REMOTE SUBMIT. In this example, */

/* the “permno.txt” file just contains a few permnos:    */

/*
12490                                              */

/*
11081                                              */

/*
10107                                              */

/* ----------------------------------------------------- */

filename pcfile ‘c:\sas\permno.txt’;

libname locallib ‘c:\sas’;

signon username=_prompt_;

rsubmit;

options linesize=72 nocenter nodate; 

libname remlib '/wrds/crsp/sasdata’;

filename unixfile ‘~/permno.txt’;

proc upload infile=pcfile outfile=unixfile;

run;

data temp;

infile unixfile;

input permno;

run;

proc sql;

  create table demo as

  select msf.permno, msf.date, msf.prc, msf.ret

  from temp, remlib.msf

  where temp.permno = msf.permno;

quit;

proc print data=demo noobs;

 where year(date) = 2001;

run;

proc download data=demo out=locallib.demo;

run;

endrsubmit;

signoff;

